Day 2 lecture

1 Coordinate Bethe Ansatz

We will now use Mathematica to study Coordinate Bethe ansatz applied to the study of the Heisemberg spin chain Hamiltonian

$$H = \sum_{n=1}^{L} (1 - P_{n,n+1}) \tag{1.1}$$

which will act on spin chain states with L spins

$$|\Psi\rangle = |\downarrow\downarrow\downarrow\uparrow \dots \downarrow\rangle + \dots \tag{1.2}$$

The permutation operator $P_{n,n+1}$ simply exchanges the position of the two neighboring sites n and $n+1$ so that for example

$$P_{12}|ab \dots\rangle = |ba \dots\rangle \tag{1.3}$$

where $a, b = \uparrow, \downarrow$. This problem is relevant for many reasons. First of all, it is a very important model of $1D$ metals first studied by Hans Bethe in 1931 [**?**]. Furthermore, related to the topic of this school, this Hamiltonian is precisely the 1-loop dilatation operator of $\mathcal{N} = 4$ SYM for operators of the form

$$\mathcal{O}_{|\Psi\rangle} = \mathrm{Tr}\,(ZZZX \dots Z) + \dots \tag{1.4}$$

where X and Z are two complex scalars of the theory, see Matthias Staudacher lecture. More precisely

$$D_{\mathcal{N}=4\,,SU(2)\,sector} = 2g^2 H\,. \tag{1.5}$$

1.1 Notation in Mathematica

We will need to introduce a good notation for the states (1.2). We will denote the state $|\downarrow\downarrow\downarrow\uparrow\rangle$ by

```
s[0,0,0,1]
```

and a more complicated state such as $|probe\rangle = |\downarrow\downarrow\downarrow\uparrow\rangle + 2|\downarrow\downarrow\uparrow\downarrow\rangle$ will correspond to

```
probe=s[0,0,0,1]+2s[0,0,1,0]
```

We can now implement the action of the Hamiltonian on these states. First we will consider the action of a single permutation operator P_{mn} which simply exchanges the spins at positions n and m. More precisely, when acting on a given state, the permutation operator **finds** all kets, represented by `s[__]` and **replaces** the n-th and $m-th$ entry in this ket

```
P[m_,n_][S_]:=S/.C_s:>ReplacePart[C,{m->C[[n]],n->C[[m]]}]
```

It is instructive to run the action of the permutator on some states such as e.g.,

```
P[1,2][probe]
P[3,4][probe]
```

We see that, as expected, the first command does nothing to the state while the second acts as it should. Now we construct the Hamiltonian acting on a state of length L. The action of the Hamiltonian then reads

```
H[L_][S_]:= L S-Sum[P[a,a+1][S],{a,1,L-1}]-P[L,1][S]
```

We can start by checking that the ground state

```
GroundState=s[0,0,0,0]
```

has zero energy by running

```
H[4][GroundState]
```

Next we can move to a more complicated state such as

```
KonishiState=s[1,1,0,0]-s[1,0,1,0]
```

This state does not transform trivially under the shift operator and thus it is not an eigenvalue on the Hamiltonian. It is thus useful to teach Mathematica that two kets are equal if they are of the form `s[list1]` and `s[list2]` with the two lists being identical up to cyclic translations. This can be implemented by

```
CanOrder[X_]:=X/.C_s:>Sort[Table[RotateLeft[C,k],{k,Length[C]}]][[1]];
```

which we can see at work running

```
CanOrder[s[0,0,1,0]]
CanOrder[s[0,0,0,1]]
```

Now we run

```
KonishiState//CanOrder
H[4][KonishiState]//CanOrder//Simplify
```

We see that this state is an eigenvalue of the Heisemberg Hamiltonian with eigenvalue 6 so that, from (1.5), we see that the one-loop anomalous dimension of the Konishi operator $\mathrm{Tr}\,[Z,X]^2$ is given by $12g^2$.

1.2 Coordinate Bethe Ansatz

We can now act with out Hamiltonian on funnier states, the so called magnon excitations – single spin flips in a ferromagnetic vacuum. To study such states is is useful to introduce the notation

```
f[list_,L_]:=s[Sequence@@Table[If[MemberQ[list,j],1,0],{j,L}]]
```

which creates a ket with spin flips at the positions in the list in a chain of size L. To see it in action run e.g.

```
f[{2,5},8]
```

which generates the state $S_2^+ S_5^+ | \downarrow\downarrow\downarrow\downarrow\downarrow\downarrow\downarrow\downarrow\rangle = | \downarrow\uparrow\downarrow\downarrow\uparrow\downarrow\downarrow\downarrow\rangle$. We will also set $L = 9$ because for what we will be interested in it is more than enough,

```
L0=9;
```

Then a single spin flip state $|F_1\rangle = \sum_{n=1}^{L} g(n) S_n^+ | \downarrow \dots \downarrow\rangle$ is given by

```
Sum[g[n] f[{n},L0],{n,L0}]
```

Now we want to check that if $g(n) = e^{ikn}$ then this state is an eigenstate of our Hamiltonian with energy $e(k)$. This is the so called single magnon state with momentum k. That is we define

```
F1=Sum[Exp[i k n] f[{n},L0],{n,L0}]
```

Next we compute

```
simpi[S_]:=Collect[S,C_s,Simplify];
```

```
h1=H[L0][F1]-e[k] F1//simpi
```

(It is instructive to remove the `//simpi` function – which collects the prefactor of each ket and simplifies it – and run this command again to understand how useful this simplification is.) From the output we get it is clear that the dispersion relation of the magnon with momentum k is simply given by

```
disp[k_]:=4Sin[k/2]^2
```

To be sure we run

```
h1/.e[k_]->disp[k]//Simplify
```

which kills most kets and simply yields the result (converting it to equation style and recognizing the 9 in the exponent as L_0)

$$(-1 + e^{ikL_0}) \left(e^{ik} | \downarrow\downarrow \dots \downarrow\downarrow\uparrow\rangle - | \uparrow\downarrow\downarrow \dots \downarrow\downarrow\rangle \right) \tag{1.6}$$

which is zero due to the periodicity condition $L_0 \times (\text{total momentum}) = L_0 k = 2\pi n$ due to the periodic nature of the ring.

Next we will consider two magnon states,

$$|F_2\rangle = \sum_{m>n}^{L} \left(e^{ikm+ipn} + S(k,p) e^{ikn+ipm} \right) S_n^+ S_m^+ | \downarrow \dots \downarrow\rangle \tag{1.7}$$

which can be inserted as

```
F2=Sum[(Exp[I k m+I p n]+S[k,p]Exp[I p m +I k n])f[{n,m},L0],{n,L0},{m,n+1,L0}];
```

First we want to compute the S-matrix $S(k,p)$ which controls the scattering of the two magnons. For that we do not want to consider the boundary terms such as the ones appearing in (1.6). Those will be related with the periodicity of the wave function and will be considered latter (actually these can trivially treated analytically). To kill them in a first stage we introduce the function

```
KillB[S_]:=S/.C_s:>0/;First[C]==1\[Or]Last[C]==1
```

which kills the kets where either the first or the last spins are up. For example

```
h1/.e[k_]->disp[k]//Simplify//KillB
```

yields zero as output. For the two magnon state

```
h2=H[L0][F2]-(disp[k]+disp[p])F2//KillB//simpi
```

As output we get several kets for which two neighbor spins are up (all other kets were already killed by our choice of plane waves as ansatz). The prefactor of each of this terms is the same up to a trivial factor so we will simply pick one of them,

```
oneothem=Coefficient[h2,f[{3,4},L0]]
```

and find the S-matrix by equating it to zero,

```
\[GothicCapitalS][k_,p_]=S[k,p]/.Solve[0==oneofthem,S[k,p]][[1]]//FullSimplify
```

We can indeed check that

```
h2/.S->\[GothicCapitalS]//Simplify
```

vanishes. Notice also that the S-matrix takes a particularly nice form under the parametrization

$$e^{ip} = \frac{u+i/2}{u-i/2} \ , \ u = \frac{1}{2}\cot\frac{p}{2} \tag{1.8}$$

as can be checked from

```
\[GothicCapitalS][k,p]/.{p->Log[(u+I/2)/(u-I/2)]/I,k->Log[(v+I/2)/(v-I/2)]/I}//Simplify
```

which simply yields

$$S(v,u) = \frac{u-v-i}{u-v+i} \,. \tag{1.9}$$

We will deal with the boundary terms latter, for now let us move to the three magnon case where integrability starts emerging in a beautiful way. The Bethe Ansatz for the many magnon wave function consists of a superposition of plane waves whose prefactors – which tell us who each magnon if scattered with the other magnons – are fixed assuming the magnons only scatter in a pairwise fashion. The function

```
BetheAnsatz[M_]:=Block[{p=Permutations[Table[j,{j,M}]]},
Sum[A[p[[i]]] Exp[Sum[I k[p[[i,j]]] n[j], {j,M}]],{i,Length[p]}]]//.
A[{a___,b_,c_,d___}]:>A[{a,c,b,d}]S[k[b],k[c]]/;b>c/.A[__]->1
```

yields the wave function $g(n_1, n_2, \ldots, n_M)$ of the M-magnon wave function

$$|F_M\rangle = \sum_{n_1<n_2<\cdots<n_M}^{L} g(n_1, \ldots, n_M) S^+_{n_1} \ldots S^+_{n_M} |\downarrow \ldots \downarrow\rangle \tag{1.10}$$

Running this function for $M \leq 4$ is very instructive to immediately see the general pattern:

```
BetheAnsatz[1]
BetheAnsatz[2]
BetheAnsatz[3]
BetheAnsatz[4]
```

The many particle state (1.10) for this wave function can be introduced as

```
F[M_]:=Block[{n},n[0]=0;Sum[BetheAnsatz[M]f[Table[n[j],{j,M}],L0],
Sequence@@Table[{n[j],n[j-1]+1,L0},{j,M}]//Evaluate]]
```

Notice that this is a fancy way to have a single function for a generic M (hence the last part of the command `Sequence@@....` It is of course possible to write in a much simpler way the three magnon wave function - it suffices to do as before for the two magnon case. Before moving on we check that indeed this general ansatz reproduces the single and double magnon case by running

```
F[1]-F1/.{k[1]->k}
F[2]-F2/.{k[1]->p,k[2]->k}
```

and seeing that these indeed vanishes. No we can check that this remarkable ansatz is indeed an eigenstate of the Hamiltonian with energy $E = \sum_j^M 4\sin^2\frac{k_j}{2}$. We consider the 3 magnon case only since the other cases can be checked similarly and all the non-trivialities are already contained in this case. Furthermore we continue to ignore the boundary terms for now. We run

```
h3=H[L0][F[3]]-Sum[disp[k[j]],{j,3}]F[3]//KillB//simpi
```

This still gives a huge amount of kets, although all kets where the spins are well separated already dropped out. Next we replace the S-matrices in this output by the two body S-matrix previously computed to obtain a remarkable zero:

```
h3/.S->\[GothicCapitalS]//Simplify
```

Now we can move to the boundary terms, i.e. consider the periodicity of the wave functions,

1.3 Bethe Equations

It is easy to see that the wave functions we considered in the previous section are periodic if $g(n_1 + L, n_2, \ldots, n_M) = g(n_2, \ldots, n_M, n_1)$ which implies simply

$$e^{ik_j L} \prod_{i \neq j}^{M} S(k_j, k_i) = 1 \tag{1.11}$$

or, in the u parametrization introduces above,

$$\left(\frac{u_j + i/2}{u_j - i/2} \right)^L = \prod_{i \neq j}^{M} \frac{u_j - u_i + i}{u_j - u_i - i} \tag{1.12}$$

Having solved these so called Bethe Ansatz equations we can simply compute the energy of the state from

$$E = \sum_{j=1}^{M} 4 \sin^2 \frac{k_j}{2} = \sum_{j=1}^{M} \frac{1}{u_j^2 + \frac{1}{4}} \tag{1.13}$$

A new very interesting story starts here when we consider the study of the solutions to the Bethe equations (1.12), see the exercises for more examples.